



# Gelato Ops Audit Report

Aug 16, 2022





# Table of Contents

<b>Summary</b>	<b>2</b>
<b>Overview</b>	<b>3</b>
<b>Issues</b>	<b>4</b>
[WP-L1] Obsoleted variables and methods after the onlyOneProxy change	4
[WP-L2] Implementation should be whitelisted to safeguard users from hijacking attack	9
[WP-L3] Empty implementation of TaskModuleBase#preCreateTask() is error-prone	10
[WP-I4] OpsProxyFactory.notProxy(owner) can be bypassed	13
[WP-I5] The restriction of only the owner of the OpsProxy can create a task that calls the OpsProxy can be bypassed	15
<b>Appendix</b>	<b>17</b>
<b>Disclaimer</b>	<b>18</b>



## Summary

This report has been prepared for Gelato Ops Audit Report smart contract, to discover issues and vulnerabilities in the source code of their Smart Contract as well as any contract dependencies that were not part of an officially recognized library. A comprehensive examination has been performed, utilizing Static Analysis and Manual Review techniques.

The auditing process pays special attention to the following considerations:

- Testing the smart contracts against both common and uncommon attack vectors.
- Assessing the codebase to ensure compliance with current best practices and industry standards.
- Ensuring contract logic meets the specifications and intentions of the client.
- Cross referencing contract structure and implementation against similar smart contracts produced by industry leaders.
- Thorough line-by-line manual review of the entire codebase by industry experts.



# Overview

## Project Summary

Project Name	Gelato Ops
Codebase	<a href="https://github.com/gelatodigital/ops">https://github.com/gelatodigital/ops</a>
Commit	8722d188b367f8bdeac6b6daadae7a834f12cb5b
Language	Solidity

## Audit Summary

Delivery Date	Aug 16, 2022
Audit Methodology	Static Analysis, Manual Review
Total Issues	5



## [WP-L1] Obsoleted variables and methods after the onlyOneProxy change

Low

### Issue Description

1. `_proxies[proxy]` can be replaced with `_ownerOf[proxy]` :

<https://github.com/gelatodigital/ops/blob/5524495a6864c51fc6479b04800278977a2e0373/contracts/opsProxy/OpsProxyFactory.sol#L87-L103>

```
88  function deployFor(address owner)
89      public
90      override
91      onlyOneProxy(owner)
92      notProxy(owner)
93      returns (address payable proxy)
94  {
95      (bytes32 seed, bytes32 salt) = _getSeedAndSalt(owner);
96
97      bytes memory bytecode = _getBytecode(owner);
98
99      proxy = _deploy(salt, bytecode);
100
101      _proxies[proxy] = true;
102      _proxyOf[owner] = proxy;
103      _ownerOf[proxy] = owner;
```

L101 can be removed.

<https://github.com/gelatodigital/ops/blob/5524495a6864c51fc6479b04800278977a2e0373/contracts/opsProxy/OpsProxyFactory.sol#L139-L141>

```
139  function isProxy(address proxy) public view override returns (bool) {
140      return _proxies[proxy];
141  }
```



`isProxy()` can be changed to:

```
139     function isProxy(address proxy) public view override returns (bool) {
140         return _ownerOf[proxy] != address(0);
141     }
```

`opsProxyFactory.isProxy(account)` can be replaced with `opsProxyFactory._ownerOf(account)` :

<https://github.com/gelatodigital/ops/blob/5524495a6864c51fc6479b04800278977a2e0373/contracts/taskModules/ProxyModule.sol#L37-L69>

```
37     function preCreateTask(address _taskCreator, address _execAddress)
38         external
39         view
40         override
41         returns (address, address)
42     {
43         bool isExecAddressProxy = opsProxyFactory.isProxy(_execAddress);
44
45         if (isExecAddressProxy) {
46             address ownerOfExecAddress = opsProxyFactory.getOwnerOf(
47                 _execAddress
48             );
49             require(
50                 _taskCreator == ownerOfExecAddress ||
51                 _taskCreator == _execAddress,
52                 "ProxyModule: Only owner of proxy"
53             );
54
55             return (ownerOfExecAddress, _execAddress);
56         } else {
57             bool isTaskCreatorProxy = opsProxyFactory.isProxy(_taskCreator);
58
59             if (isTaskCreatorProxy) {
60                 address ownerOfTaskCreator = opsProxyFactory.getOwnerOf(
61                     _taskCreator
62                 );
63
64                 return (ownerOfTaskCreator, _execAddress);
65             }
66         }
```



```
67         return (_taskCreator, _execAddress);
68     }
69 }
```

## Recommendation

1. Remove `getOwnerOf()` and rename `_ownerOf` to `ownerOf` and make it `public` ;
2. Change `preCreateTask()` to:

```
37 function preCreateTask(address _taskCreator, address _execAddress)
38     external
39     view
40     override
41     returns (address, address)
42 {
43     address ownerOfExecAddress = opsProxyFactory.ownerOf(
44         _execAddress
45     );
46
47     if (ownerOfExecAddress != address(0)) {
48         require(
49             _taskCreator == ownerOfExecAddress ||
50             _taskCreator == _execAddress,
51             "ProxyModule: Only owner of proxy"
52         );
53
54         return (ownerOfExecAddress, _execAddress);
55     } else {
56         address ownerOfTaskCreator = opsProxyFactory.ownerOf(
57             _taskCreator
58         );
59
60         if (ownerOfTaskCreator != address(0)) {
61             return (ownerOfTaskCreator, _execAddress);
62         }
63
64         return (_taskCreator, _execAddress);
65     }
66 }
```



## 2. `_nextSeeds` is no longer needed:

<https://github.com/gelatodigital/ops/blob/5524495a6864c51fc6479b04800278977a2e0373/contracts/opsProxy/OpsProxyFactory.sol#L155-L163>

```
155  function _getSeedAndSalt(address _account)
156      internal
157      view
158      returns (bytes32 seed, bytes32 salt)
159  {
160      seed = _nextSeeds[_account];
161
162      salt = keccak256(abi.encode(_account, seed));
163  }
```

`seed` can only be `0` and `salt` will always be `keccak256(abi.encode(_account, 0))` as only one proxy is allowed for one `account`.

Therefore, `deployFor()` can be changed to:

<https://github.com/gelatodigital/ops/blob/5524495a6864c51fc6479b04800278977a2e0373/contracts/opsProxy/OpsProxyFactory.sol#L88-L110>

```
88  function deployFor(address owner)
89      public
90      override
91      onlyOneProxy(owner)
92      notProxy(owner)
93      returns (address payable proxy)
94  {
95      proxy = _deploy(bytes32(uint256(uint160(owner))), _getBytecode(owner));
96
97      _proxyOf[owner] = proxy;
98      _ownerOf[proxy] = owner;
99
100      emit DeployProxy(msg.sender, owner, address(proxy));
101  }
```

`determineProxyAddress()` should also be changed accordingly.





`getNextSeed()` , `_getSeedAndSalt()` can be removed.

Using `bytes32(owner)` may also be unnecessary, could just use `bytes32(0)` as the salt.

## Status

✓ Fixed



## [WP-L2] Implementation should be whitelisted to safeguard users from hijacking attack

Low

### Issue Description

The current implementation allows the owner of the `OpsProxy` to call `upgradeTo()` and upgrade to an arbitrary new implementation:

<https://github.com/gelatodigital/ops/blob/5524495a6864c51fc6479b04800278977a2e0373/contracts/vendor/proxy/EIP173/EIP173Proxy.sol#L66-L68>

```
66  function upgradeTo(address newImplementation) external onlyProxyAdmin {  
67      _setImplementation(newImplementation, "");  
68  }
```

This could be a problem if the attacker managed to hijack the frontend and deceived the user into calling `upgradeTo()` and set to a malicious implementation.

### Recommendation

Consider overriding the `upgradeTo()` and `upgradeToAndCall()`, only allows `_setImplementation` to a whitelisted implementation on `OpsProxyFactory.sol`.

### Status

✓ Fixed



## [WP-L3] Empty implementation of TaskModuleBase#preCreateTask() is error-prone

Low

### Issue Description

<https://github.com/gelatodigital/ops/blob/5524495a6864c51fc6479b04800278977a2e0373/contracts/taskModules/TaskModuleBase.sol#L12-L17>

```
12     function preCreateTask(address, address)
13         external
14         virtual
15         override
16         returns (address, address)
17     {}
```

The default empty implementation will return `address(0), address(0)` as the `taskCreator` and `execAddress`.

If configured correctly, the empty implementation of `preCreateTask()` will not be called unless `module.requirePreCreate() == true`:

<https://github.com/gelatodigital/ops/blob/5524495a6864c51fc6479b04800278977a2e0373/contracts/libraries/LibTaskModule.sol#L23-L56>

```
23     function preCreateTask(
24         address _taskCreator,
25         address _execAddress,
26         mapping(LibDataTypes.Module => address) storage taskModuleAddresses
27     ) internal returns (address, address) {
28         uint256 length = uint256(type(LibDataTypes.Module).max);
29
30         for (uint256 i; i <= length; i++) {
31             LibDataTypes.Module module = LibDataTypes.Module(i);
32             if (!module.requirePreCreate()) continue;
33
34             address moduleAddress = taskModuleAddresses[module];
35             _moduleInitialised(moduleAddress);
```



```
36
37     bytes memory delegatecallData = abi.encodeWithSelector(
38         ITaskModule.preCreateTask.selector,
39         _taskCreator,
40         _execAddress
41     );
42
43     (, bytes memory returnData) = _delegateCall(
44         moduleAddress,
45         delegatecallData,
46         "Ops.preCreateTask: "
47     );
48
49     (_taskCreator, _execAddress) = abi.decode(
50         returnData,
51         (address, address)
52     );
53 }
54
55 return (_taskCreator, _execAddress);
56 }
```

<https://github.com/gelatodigital/ops/blob/5524495a6864c51fc6479b04800278977a2e0373/contracts/Ops.sol#L46-L67>

```
46     function createTask(
47         address _execAddress,
48         bytes calldata _execDataOrSelector,
49         LibDataTypes.ModuleData calldata _moduleData,
50         address _feeToken
51     ) external override returns (bytes32 taskId) {
52         address taskCreator;
53
54         (taskCreator, _execAddress) = LibTaskModule.preCreateTask(
55             msg.sender,
56             _execAddress,
57             taskModuleAddresses
58         );
59
60         taskId = _createTask(
61             taskCreator,
```



```
62     _execAddress,  
63     _execDataOrSelector,  
64     _moduleData,  
65     _feeToken  
66 );  
67 }
```

We believe it will be less error-prone if it reverts instead of returning `address(0), address(0)` .

## Recommendation

Consider changing to:

```
12 function preCreateTask(address, address)  
13     external  
14     virtual  
15     override  
16     returns (address, address)  
17 {  
18     revert("Not Implemented");  
19 }
```

## Status

✓ Fixed



## [WP-I4] OpsProxyFactory.notProxy(owner) can be bypassed

### Informational

### Issue Description

While there is a `notProxy(owner)` modifier on `deployFor()`, this restriction can be bypassed by creating the proxy of a proxy before creating that proxy:

1. Query the `determineProxyAddress()` for the `account` ;
2. `deployFor()` with the address above as the `owner` ;
3. `deployFor()` with `account` as the `owner` .

This may or may not be a problem depending on what `notProxy(owner)` was designed for.

<https://github.com/gelatodigital/ops/blob/5524495a6864c51fc6479b04800278977a2e0373/contracts/opsProxy/OpsProxyFactory.sol#L88-L110>

```
88  function deployFor(address owner)
89      public
90      override
91      onlyOneProxy(owner)
92      notProxy(owner)
93      returns (address payable proxy)
94  {
95      (bytes32 seed, bytes32 salt) = _getSeedAndSalt(owner);
96
97      bytes memory bytecode = _getBytecode(owner);
98
99      proxy = _deploy(salt, bytecode);
100
101      _proxies[proxy] = true;
102      _proxyOf[owner] = proxy;
103      _ownerOf[proxy] = owner;
104
105      unchecked {
106          _nextSeeds[owner] = bytes32(uint256(seed) + 1);
107      }
108
109      emit DeployProxy(msg.sender, owner, seed, salt, address(proxy));
110  }
```



<https://github.com/gelatodigital/ops/blob/5524495a6864c51fc6479b04800278977a2e0373/contracts/opsProxy/OpsProxyFactory.sol#L35-L38>

```
35  modifier notProxy(address _account) {  
36      require(!isProxy(_account), "OpsProxyFactory: No proxy");  
37      _;  
38  }
```

<https://github.com/gelatodigital/ops/blob/5524495a6864c51fc6479b04800278977a2e0373/contracts/opsProxy/OpsProxyFactory.sol#L139-L141>

```
139  function isProxy(address proxy) public view override returns (bool) {  
140      return _proxies[proxy];  
141  }
```

## Status

 Acknowledged



## [WP-I5] The restriction of only the owner of the OpsProxy can create a task that calls the OpsProxy can be bypassed

### Informational

### Issue Description

<https://github.com/gelatodigital/ops/blob/5524495a6864c51fc6479b04800278977a2e0373/contracts/taskModules/ProxyModule.sol#L37-L69>

```
37  function preCreateTask(address _taskCreator, address _execAddress)
38      external
39      view
40      override
41      returns (address, address)
42  {
43      bool isExecAddressProxy = opsProxyFactory.isProxy(_execAddress);
44
45      if (isExecAddressProxy) {
46          address ownerOfExecAddress = opsProxyFactory.getOwnerOf(
47              _execAddress
48          );
49          require(
50              _taskCreator == ownerOfExecAddress ||
51              _taskCreator == _execAddress,
52              "ProxyModule: Only owner of proxy"
53          );
54
55          return (ownerOfExecAddress, _execAddress);
56      } else {
57          bool isTaskCreatorProxy = opsProxyFactory.isProxy(_taskCreator);
58
59          if (isTaskCreatorProxy) {
60              address ownerOfTaskCreator = opsProxyFactory.getOwnerOf(
61                  _taskCreator
62              );
63
64              return (ownerOfTaskCreator, _execAddress);
65          }
66
67          return (_taskCreator, _execAddress);
```





```
68     }  
69 }
```

By front-running a proxy creating transaction ( `deployFor()` ) and create a task for that address, which is not yet deployed, this restriction can be bypassed.

Because the `OpsProxyFactory._proxies[proxy]` is not yet set to `true` , it will not go into the branch of L45-56.

Thanks to the check on OpsProxy, we believe there is no way to exploit it though:

<https://github.com/gelatodigital/ops/blob/5524495a6864c51fc6479b04800278977a2e0373/contracts/opsProxy/OpsProxy.sol#L16-L31>

```
16  modifier onlyAuth() {  
17      require(  
18          msg.sender == ops || msg.sender == owner(),  
19          "OpsProxy: Not authorised"  
20      );  
21  
22      if (msg.sender == ops) {  
23          address taskCreator = _getTaskCreator();  
24  
25          require(  
26              taskCreator == owner(),  
27              "OpsProxy: Only tasks created by owner"  
28          );  
29      }  
30      _;  
31  }
```

## Status

ⓘ Acknowledged



# Appendix

## Timeliness of content

The content contained in the report is current as of the date appearing on the report and is subject to change without notice, unless indicated otherwise by WatchPug; however, WatchPug does not guarantee or warrant the accuracy, timeliness, or completeness of any report you access using the internet or other means, and assumes no obligation to update any information following publication.



## Disclaimer

This report is based on the scope of materials and documentation provided for a limited review at the time provided. Results may not be complete nor inclusive of all vulnerabilities. The review and this report are provided on an as-is, where-is, and as-available basis. You agree that your access and/or use, including but not limited to any associated services, products, protocols, platforms, content, and materials, will be at your sole risk. Smart Contract technology remains under development and is subject to unknown risks and flaws. The review does not extend to the compiler layer, or any other areas beyond the programming language, or other programming aspects that could present security risks. A report does not indicate the endorsement of any particular project or team, nor guarantee its security. No third party should rely on the reports in any way, including for the purpose of making any decisions to buy or sell a product, service or any other asset. To the fullest extent permitted by law, we disclaim all warranties, expressed or implied, in connection with this report, its content, and the related services and products and your use thereof, including, without limitation, the implied warranties of merchantability, fitness for a particular purpose, and non-infringement. We do not warrant, endorse, guarantee, or assume responsibility for any product or service advertised or offered by a third party through the product, any open source or third-party software, code, libraries, materials, or information linked to, called by, referenced by or accessible through the report, its content, and the related services and products, any hyperlinked websites, any websites or mobile applications appearing on any advertising, and we will not be a party to or in any way be responsible for monitoring any transaction between you and any third-party providers of products or services. As with the purchase or use of a product or service through any medium or in any environment, you should use your best judgment and exercise caution where appropriate. FOR AVOIDANCE OF DOUBT, THE REPORT, ITS CONTENT, ACCESS, AND/OR USAGE THEREOF, INCLUDING ANY ASSOCIATED SERVICES OR MATERIALS, SHALL NOT BE CONSIDERED OR RELIED UPON AS ANY FORM OF FINANCIAL, INVESTMENT, TAX, LEGAL, REGULATORY, OR OTHER ADVICE.